How to clean Enhanced TRACE data

Jens Dick-Nielsen *

December 3, 2014

Abstract

This note is a supplement to "Liquidity biases in TRACE" from the Journal of Fixed Income, 2009. The note updates the transaction data filter to handle Enhanced Historic TRACE data and provides SAS code for the new filter (now including the 2012 change). As an example, there are 6.7 mill raw transaction reports in 2007. 440,000 are deleted as known errors. 780,000 are deleted as agency transactions. Finally, 1,6 mill are deleted as interdealer double counted transactions. In total the filter deletes almost 35% of the raw transactions.

Keywords: Enhanced TRACE data; Error filter; SAS code.

JEL: G10; G12

* Department of Finance, Copenhagen Business School, Solbjerg Plads 3, DK-2000 Frederiksberg, Denmark (e-mail: jdn.fi@cbs.dk). All comments and questions are welcome. This version contains an updated version of the filter. Thanks to Philipp Schuster, Karlsruhe Institute of Technology (KIT), for pointing out programming errors and suggested solutions in the earlier filter, and for drawing my attention to the 2012 change in TRACE. All remaining errors are my own.

Introduction

The Enhanced Historic TRACE corporate bond data is an alternative to the standard TRACE corporate bond data. The enhanced data contains transaction reports for all transactions back to the inception of TRACE in July, 2002. This includes transactions in formerly non-disseminated bonds (except 144A bonds). The enhanced data furthermore contains uncapped transaction volumes and historical buy-sell side information as the most significant improvements over the standard data. Finally, the enhanced data contains some more specialized information which among other things allows for a better error filtering algorithm (eg. information on reporting date and time).

The enhanced information comes at a cost of an 18 month lag in availability of the data. Hence, the standard TRACE data will usually have 15 months more data but with less information available. Note however, that the newest standard TRACE data is only missing the uncapped volumes when compared to the enhanced data, since essentially all non-144A transactions are now disseminated and buy-sell side information has also been included in the standard TRACE data since November 2008.

This note and the filter presented have been updated as of December, 2014. The update allows the filter to handle the 2012 change in the data structure. The update also changes the way agency transactions are handled in the filter.

The following discussion assumes that the reader is familiar with the issues and terminology from Dick-Nielsen (2009).

Cleaning Enhanced Historic TRACE data

This section describes the filtering algorithm presented in Dick-Nielsen (2009). The filter is designed to delete transactions which are already known and marked as being errors. The section also describes the change in the data structure which was effectuated February 6th, 2012.

The error filter

The filter for the enhanced data runs in 3 steps much like the original filter in Dick-Nielsen (2009). However, the steps are not exactly identical to the original filter. The first two steps of the filter are necessary in order to delete the errors whereas the last step (regarding agency transactions and order special circumstances) is convenient for many applications.

Step 1: Clean same-day corrections and cancelations. Same-day refers to corrections and cancelations reported within the same reporting date (not transaction date). These can be uniquely identified by the link between the message sequence number and the original message sequence number. The message sequence number is unique on an intra-reporting day level. Note that reporting date and transaction date need not be the same.

Step 2: Remove reversals and the matching original transaction report. Reversals are cancelations reported on a later date than the date on which the original transaction took place (transaction date $<$ reporting date). Each reversal should be linked to exactly one original transaction. If for some reason more than one transaction can be linked (by matching) to the same

reversal, only one of the matching reports is deleted. The information in the reversal report should exactly match the original report which it is reversing except that the As-Of indicator for the reversal contains an 'R' and the reporting date for the reversal lies after the reporting date for the original report (or else it would not be a reversal but a same-day cancelation).

Step 3: This step deletes agency transactions (see discussion in a later section) where the principal transaction has the same price as the agency transaction (a sort of double counting). This step furthermore deletes one of the reports in each interdealer transaction pair (not an issue with standard TRACE data) and classifies the retained report as an interdealer transaction. Finally, this step could delete special transactions such as commissioned transactions, odd number of days to settlement, special price flag indicated etc.

Note that except for the double counting of interdealer transactions, the transactions deleted in step 3 could as well have been kept in the sample since they are not actual errors. However, it is common in the literature to delete at least some of these transactions as being potential outliers.

The 2012 change in data structure

In February 2012 the data structure in the TRACE database was changed. The change makes the enhanced filtering simpler. Prior to the change any corrections made on a later date was a reversal. After the change any change made after T-21 days is a reversal. The time span for what used to be 'same-day' corrections have thus increased. Specifically, the coding of the variable

Trade Status has been changed. The overall filtering algorithm is still as outlined above. Below are examples which illustrate the effect of the data structure change.

Table 1 shows an example of a cancelation made after the 2012 change. The first transaction is the original transaction. The second transaction marked with an 'X' in the Trade Status is the cancelation report. The cancelation will have the exact same message sequence number as the report it is canceling. All information in the cancelation will match that of the reversal except for the trade status, the reporting date, and time. The first two reports in the table contain all the relevant information and should both be deleted as errors. The third transaction in the table could be the correct transaction, however note that it is not necessary to identify any such third transaction in the filtering.

Table 2 shows an example of a new correction made after the 2012 change. The first transaction is the original report with the error. The second report is identical to the first report but with a 'C' in the Trade Status (and possibly another reporting date and time). Note that the message sequence number matches that of the original transaction. The first two reports with identical message sequence numbers should both be deleted as errors. The third transaction is the replacement report or correction. The replacement report contains all the correct information and have an 'R' in the trade status indicator. For the replacement report the original message sequence number will match that of the original transaction.

Table 3 shows an example of a reversal. After the 2012 change reversals have become very rare events. In the table the first transaction is the original

transaction. The third transaction is identical to the first except for the 'Y' in the trade status indicator (and the 'R' in the reversal and the reporting date and time). This time the original message sequence number shown for the reversal matches that of the original report. Both the original report and the reversal should be deleted.

SAS code implementing the filter for the enhanced TRACE data is presented in the last section of this note. The code can be directly copied into SAS.

Differences from the original filter

The filter described above for cleaning the Enhanced Historic TRACE data is not identical to the filter in Dick-Nielsen (2009). The method is, however, more or less the same and the identification of error reports is still as in Dick-Nielsen (2009). The following things have been changed (for the filter that handles data prior to 2012):

A) Step 1 in the original filter deleted true duplicates in the TRACE data. Compared to the other steps, this step deleted very few observations. Dick-Nielsen (2009) noted that there should be no true duplicates in the data; hence the step should have been unnecessary. The reports deleted in this step are actually not true duplicates. The duplicates were identified by cusip, transaction date, and message sequence number. However, the message sequence number is unique within the reporting date, not the transaction date. The "duplicates" arise when an As-Of transaction in a specific bond happens

to get the same message sequence number as one of the timely reported transactions in the same specific bond. Each "duplicate" cluster thus consists of one timely reported transaction and a number of transactions reported on a later date. This is a rather unlikely event and explains why there were very few observations deleted in this step in the original filter. Since the reporting date is known in the Enhanced TRACE data, the message sequence number is in fact unique within that reporting date.

B) The original filter first deletes reversals and then deletes same-day corrections and cancelations. The filter presented above does it the other way around. The reasoning in Dick-Nielsen (2009) was that reversals are chronologically newest and as such should be considered first. While this reasoning is not incorrect, it is also possible to give another argument. The key example is a same-day correction with a reversal matching the corrected transaction. This involves three transaction reports; 1 original report, 1 correction report (same-day), and 1 reversal report (filed on a later date). In the original filter, the reversal would delete the correction report and the original report would remain in the data sample. In the new filter the same-day correction deletes the original transaction (it replaces it) and then the reversal deletes the corrected report. This results in the deletion of all three reports. It is not possible to rule out one of the two scenarios as being incorrect, however the latter scenario is chosen for the new filter as the most plausible. The situation is either way very uncommon, so it does not make a huge difference for the filtering.

C) The new filter deletes special transactions etc. as the last step instead

of as an initial step. This is chosen so that same-day corrections and reversals can be matched with original transactions in the best possible way. If the filter deletes some transactions initially then it can be a problem when matching transactions later in the process. The new approach reduces the number of unmatched reversals, but it does not make a significant difference for the final and clean data sample.

D) The original filter does not allow a reversal to cancel an As-Of transaction marked with an 'A' since it is not possible to know which report came first in a chronological sense. The reversal could have been reported before the As-Of 'A' transaction in which case it could not be canceling the 'A' transaction. This is no longer an issue since the Enhanced TRACE data contain information on reporting time and date. However, as with the above issues, this does not make a significant difference for the final and clean data sample.

Agency transactions

This version of the filter takes a slightly new approach to agency transactions. Agency transactions are not errors and could be kept in the data. Hence, deleting agency transactions is a choice. In the filter presented in the last section agency transactions are deleted if they are dealer-customer transactions without commission.

The typical agency transaction consists of an introducing broker acting on behalf of a customer by transacting with an executing broker. The intro-

ducing broker has to pass on the bond at the same price as that taken by the executing broker. However, the introducing broker can charge a commission. This sequence of trading will produce three reports in TRACE. Table 4 shows an example of an agency transaction. The first transaction report is from the executing broker transacting with the introducing broker. From the perspective of the executing broker it is a principal transaction. The third transaction report is the other side of the interdealer transaction as reported by the introducing broker. From the perspective of the introducing broker it is an agency transaction. Finally, the second transaction report is when the introducing broker transacts with the customer. This transaction is also an agency transaction. Note that one can only trust the seller/buyer capacity when it is the reporting party that reports it. In the first transaction report the executing broker reports that both seller and buyer capacity is a principal transaction. This is not correct since the counterparty broker is acting as an introducing broker and from that perspective it is an agency transaction.

In the table 4 the dealer-customer transaction price is reported with a commission. This is not always the case since there is sometimes no commission charged in agency transactions. When there is no commission then all three prices will be the same. However, this is not correct since the customer will most likely pay an additional cost, just not in the form of a commission e.g. a wrap fee. The price sequence is thus not reflective of the actual costs of transacting for the customer when there is no commission. For this reason the filter in the last section deletes dealer-customer transactions without commission.

Table 5 shows a rough estimate of the price levels in agency transactions

compared to principal transactions. Agency transactions with commissions are largely comparable to principal customer transaction. The table is estimated using intra-day intra-bond differences between principal interdealer transactions and interdealer agency transaction, between agency customer transactions (with and without commission) and principal customer transactions. All estimation is done for transactions with volume below $100,000, because agency transactions larger than this are unusual. The reference point of 100 for interdealer transactions has been selected for illustration and then the differences are add/subtracted. What the table calls 'agency sell/buy' is the interdealer prices from table 4. Hence, when the introducing broker negotiates with the executing broker on behalf of the customer then the prices are usually slightly worse than for principal interdealer transaction with the dealer negotiating on his own behalf. However, when the possible commission is added/subtracted the final price for the customer is not that different from the customer principal transaction.

Price sequence filter

The filter presented here and the filters in Dick-Nielsen (2009) are designed to clean for known errors. These procedures cannot be replaced by a price sequence filter seeking to delete outliers as was discussed in Dick-Nielsen (2009). Even though there should not be any typos in the TRACE data as discussed in Dick-Nielsen (2009), it can still be useful to subsequently apply a price sequence filter in order to avoid that a few unusual observations drive overall results. Rossi (2014) contains a good description of a powerful price sequence filter for TRACE data. The SAS code for this filter can be

downloaded from the author's website.

SAS code

```
*************************************************************************
* 20141201
*
* (C) Jens Dick-Nielsen
*     Copenhagen Business School
*     jdn.fi@cbs.dk
*
*
* Description:
*      Takes an Enhanced TRACE data sample from WRDS and cleans out
*      the reporting errors. See details in the note "How to clean
*      Enhanced TRACE data". For more details on the algorithm see
*      the article "Liquidity biases in TRACE".
*
*
* NB:  The code contains some parts which are optional.
*      These are by default not enabled but may be of use (see the
*      last part of the code).
*
* Acknowledgement:
*      Special thanks to Philipp Schuster, Karlsruhe Institute of
*      Technology (KIT), for pointing out programming errors and
*      suggested solutions in the earlier filter, and for drawing
*      my attention to the 2012 change in TRACE. All remaining
*      errors are my own.
*
*************************************************************************


* The name of the original Enhanced TRACE dataset
* supplied to the program;
%LET IN = traceIN;
* The name given to the cleaned Enhanced TRACE dataset;
%Let OUT = traceCLEAN;

* Path to your library containing the dataset;
* The cleaned dataset will be stored in this location;
 Libname TRACE 'C:\Enhanced␣TRACE';
```

```
*********************************************************************
*
* Start of the program
*
*
*********************************************************************;

* Erases the log and output screens;
dm 'clear log';
dm 'clear output';


*************************
*
* POST 2012 change
*
*************************;

* Cleans data reported after Feb 6th, 2012;
* The coding and reporting structure changed with the transition
* to the TRAQS reporting system;
* Specifically, the link between a reversal and the original
* transaction is now unique and transperant;

data temp_raw
       temp_deleteI_NEW (keep = cusip_id entrd_vol_qt rptd_pr
           trd_exctn_dt trd_exctn_tm rpt_side_cd cntra_mp_id
               msg_seq_nb)
       temp_deleteII_NEW (keep = cusip_id entrd_vol_qt rptd_pr
           trd_exctn_dt trd_exctn_tm rpt_side_cd cntra_mp_id
               orig_msg_seq_nb);
 set trace.&IN;
 where trd_rpt_dt ge '06Feb2012'd;

     * Deletes observations without a cusip_id;
     if cusip_id = '' then delete;
```

```
     * Takes out all cancellations and corrections;
     * These transactions should be deleted together with the
     * original report;
     if trc_st in ('X','C') then output temp_deleteI_NEW;
     * Reversals. These have to be deteled as well together with
     * the original report;
     else if trc_st in ('Y') then output temp_deleteII_NEW;
     * The rest of the data;
     else output temp_raw;

run;

* Deletes the cancellations and corrections as identified by
* the reports in temp_deleteI_NEW;
* These transactions can be matched by message sequence number
* and date. We furthermore match on cusip, volume, price, date,
* time, buy-sell side, contra party;
* This is as suggested by the variable description;

proc sql;
 CREATE TABLE temp_raw2 AS
 select * from temp_raw as a,
 ( (select cusip_id, entrd_vol_qt, rptd_pr, trd_exctn_dt,
 trd_exctn_tm, rpt_side_cd, cntra_mp_id, msg_seq_nb from temp_raw)
 except
 (select cusip_id, entrd_vol_qt, rptd_pr, trd_exctn_dt,
    trd_exctn_tm, rpt_side_cd, cntra_mp_id, msg_seq_nb
        from temp_deleteI_NEW) ) as b
  where a.cusip_id=b.cusip_id
  and a.entrd_vol_qt=b.entrd_vol_qt
  and a.rptd_pr=b.rptd_pr
  and a.trd_exctn_dt=b.trd_exctn_dt
  and a.trd_exctn_tm= b.trd_exctn_tm
  and a.rpt_side_cd=b.rpt_side_cd
  and a.cntra_mp_id= b.cntra_mp_id
  and a.msg_seq_nb = b.msg_seq_nb;
quit;

* Deletes the reports that are matched by the reversals;
proc sql;
```

```
 CREATE TABLE temp_raw3_NEW AS
 select * from temp_raw2 as a,
 ( (select cusip_id, entrd_vol_qt, rptd_pr, trd_exctn_dt,
 trd_exctn_tm, rpt_side_cd, cntra_mp_id, msg_seq_nb from temp_raw2)
 except
 (select cusip_id, entrd_vol_qt, rptd_pr, trd_exctn_dt,
 trd_exctn_tm, rpt_side_cd, cntra_mp_id, orig_msg_seq_nb
 from temp_deleteII_NEW) ) as b
  where a.cusip_id=b.cusip_id
  and a.entrd_vol_qt=b.entrd_vol_qt
  and a.rptd_pr=b.rptd_pr
  and a.trd_exctn_dt=b.trd_exctn_dt
  and a.trd_exctn_tm= b.trd_exctn_tm
  and a.rpt_side_cd=b.rpt_side_cd
  and a.cntra_mp_id= b.cntra_mp_id
  and a.msg_seq_nb = b.msg_seq_nb;
quit;
* Ends the filtering of the post-change data;



*************************
*
* PRE 2012 change
*
*************************;

* Takes same-day corrections and splits them into two data sets;
* 1 for all the correct trades, and 1 for the corrections;
data temp_raw temp_delete (keep = TRD_RPT_DT ORIG_MSG_SEQ_NB);
 set trace.&IN;
 where trd_rpt_dt < '06Feb2012'd;

     * Deletes observations without a cusip_id;
     if cusip_id = '' then delete;

     * Takes out all cancellations into the temp_delete dataset;
     if trc_st = 'C' then output temp_delete;
     * All corrections are put into both datasets;
     else if trc_st = 'W' then output temp_delete temp_raw;
```

```
     else output temp_raw;

run;

* Deletes the error trades as identified by the message
* sequence numbers. Same day corrections and cancelations;
proc sql;
 CREATE TABLE temp_raw2 AS
 select * from temp_raw as a,
 ( (select MSG_SEQ_NB, TRD_RPT_DT from temp_raw)
 except
 (select ORIG_MSG_SEQ_NB, TRD_RPT_DT from temp_delete) ) as b
  where a.msg_seq_nb=b.msg_seq_nb and a.TRD_RPT_DT =b.TRD_RPT_DT ;
quit;

* Take out reversals into a dataset;
data reversal temp_raw3;
 set temp_raw2;
 N=_N_;
 if asof_cd='R' then output reversal;
 else output temp_raw3;
run;

* Sorting the data so that it can be merged;
proc sort data=reversal (drop = N) nodupkey; by trd_exctn_dt
cusip_id trd_exctn_tm rptd_pr entrd_vol_qt rpt_side_cd cntra_mp_id
trd_rpt_dt trd_rpt_tm MSG_SEQ_NB; run;

proc sort data=temp_raw3; by trd_exctn_dt cusip_id trd_exctn_tm
rptd_pr entrd_vol_qt rpt_side_cd cntra_mp_id; run;

* Merges reversals back on and selects matching observations;
data reversal2;
 merge temp_raw3 (in=qqq) reversal (in=qq) ;
 by trd_exctn_dt cusip_id trd_exctn_tm rptd_pr
    entrd_vol_qt rpt_side_cd cntra_mp_id;
 if qq=1;
 if qqq=1;

 * Reversal have to be on a later date
```

```
 * (or else it would not be a reversal);
 * i.e. we do not delete potential as_of trades
 * from a later date that may match;
 if trd_exctn_dt < trd_rpt_dt;

run;

* Selects only 1 matching reversal (and keeps the rest);
proc sort data=reversal2 nodupkey; by trd_exctn_dt bond_sym_id
trd_exctn_tm rptd_pr entrd_vol_qt; run;

proc sort data=reversal2; by N; run;

proc sort data=temp_raw3; by N; run;

* Deletes the macthing reversals;
data temp_raw4;
 merge reversal2 (in=qq) temp_raw3;
 by N;
 if qq=0;
run;
* Ends the filter for PRE-change data;


*************************
*
* Combines the PRE and POST data into one
*
*************************;

data temp_raw_comb (drop = N asof_cd trd_rpt_dt trd_rpt_tm
MSG_SEQ_NB trc_st orig_msg_seq_nb);
 set temp_raw4 temp_raw3_NEW;
run;


*************************
*
* Agency transaction filtering
*
```

```
*************************;

* This step deletes agency transactions but is not part
* of the error detection filter. This step can be deleted if
* you want to keep all agency transactions;

* Deletes agency customer transactions without commission;
* These transactions will have the same price as the
* interdealer transaction (if reported correctly);
data temp_raw6 (drop = agency);
 set temp_raw_comb;

 * Identifies agency transactions;
 if rpt_side_cd='B' then agency=buy_cpcty_cd;
 else if rpt_side_cd='S' then agency=sell_cpcty_cd;

 * Deletes agency transactions which are dealer-customer
 * transactions without commission;
 if agency='A' and cntra_mp_id = 'C' and CMSN_TRD = 'N' then delete;

run;


data trace.&out;
 set temp_raw6;

     * Deletes interdealer transactions (one of the sides);
     * Renames the reporting party side indicator to include
     * a D for interdealer;
     if cntra_mp_id = 'D' and rpt_side_cd='B' then delete;
     if cntra_mp_id = 'D' and rpt_side_cd='S' then rpt_side_cd='D';

     * Deletes WI trades;
       *if WIS_FL = 'N';
     * Deletes trades which are not secondary market;
       *if TRDG_MKT_CD in ('S2','P1','P2') then delete;
     * Deletes if it trades under special circumstances;
       *if SPCL_TRD_FL = 'Y' then delete;
     * Deletes if it is an equity linked note;
       *if SCRTY_TYPE_CD = 'C';
```

```
     * If days to settlement is very non-standard then
     * delete it (6 is arbitrary). From a certain date the
     * settlement date is given instead of the days to settle;
       *if DAYS_TO_STTL_CT<6;
     * Deletes if it is not a cash sale;
       *if sale_cndtn_cd = 'C' then delete;
     * Deletes commissioned trades;
       *if CMSN_TRD = 'N';
     * Deletes a trade if it is an automatic give up;
       *if AGU_QSR_ID in ('A','Q') then delete;

run;
```

References

Dick-Nielsen, J., 2009. Liquidity biases in TRACE. Journal of Fixed Income 19, 43–55.

Rossi, M., 2014. Realized Volatility, Liquidity, and Corporate Yield Spreads. Quarterly Journal of Finance 4, 1–42.

Table 1 Cancellations after Feb 6, 2012.

The table contains a typical example of how a cancelation is reported in TRACE after Feb 6, 2012. A trade can be canceled within 21 days by filing a new report identical to the target report. The cancelation report will have an 'X' in the trade status and could have been filing on a later date as in the example below. However, note that the message sequence number is identical to the target report which the filer wants to cancel. In this case it appears that the same transaction has been filed twice and that one of them furthermore has a wrong time stamp and wrong buy-sell side information. It is not necessary to identify three transactions as in this example. Only the first two reports are part of the cancelation and these two reports should be deleted. The variable in the table are bond ID (bond_sym_id), transaction date (trd_exctn_dt), transaction time (trd_exctn_tm), price (rptd_pr), par volume (entrd_vol_qt), trade status (trc_st), reporting date (trd_rpt_dt), buy-sell side (rpt_side_cd), and message sequence number (msg_seq_nb).

Bond ID	Transaction Date	Transaction Time	Price	Par volume	Report Date	Trade Status	Buy-Sell side	Msg Seq Number
AA.AA	20121107	8:51:00	100.8	50,000	20121107	T	B	0002805
AA.AA	20121107	8:51:00	100.8	50,000	20121115	X	B	0002805
AA.AA	20121107	8:54:00	100.8	50,000	20121107	T	S	0002997

Table 2 Corrections after Feb 6, 2012.

The table contains a typical example of how a correction is reported in TRACE after Feb. 6, 2012. A trade can be corrected within 21 days by filing two reports, one new report identical to the target report and a new report with the correct information. The cancelation-correction report will have a 'C' in the trade status and the new correct report will have a 'R' in the trade status. The cancelation-correction will work just like the pure cancelation report (with the 'X' in the trade status). The new correction will have an original message sequence number identical to the message sequence number of the report that is canceled by the cancelation-correction. The original report and the cancelation-correct should be deleted. The variable in the table are bond ID (bond_sym_id), transaction date (trd_exctn_dt), transaction time (trd_exctn_tm), price (rptd_pr), par volume (entrd_vol_qt), trade status (trc_st), message sequence number (msg_seq_nb) and original message sequence number (orig_msg_seq_nb).

Bond ID	Transaction Date	Transaction Time	Price	Par volume	Trade Status	Msg Seq Number	Orig Msg Seq Number
AA.AA	20121107	16:08:45	103.071	5,000,000	T	0061736	
AA.AA	20121107	16:08:45	103.071	5,000,000	C	0061736	
AA.AA	20121107	16:08:45	103.071	500,000	R	0062455	0061736

Table 3 Reversals after Feb 6, 2012.

The table contains a typical example of how a reversal is reported in TRACE after Feb. 6, 2012. A reversal report has to be filed if a transaction report is canceled (corrected) after more than 21 days. The reversal will then cancel the original transaction. In case that the original report needs to be corrected then the correction will be filed as a new as-of transaction report. The variable in the table are bond ID (bond_sym_id), transaction date (trd_exctn_dt), transaction time (trd_exctn_tm), price (rptd_pr), par volume (entrd_vol_qt), trade status (trc_st), as of (asof_cd) message sequence number (msg_seq_nb) and original message sequence number (orig_msg_seq_nb).

Bond ID	Transaction Date	Transaction Time	Price	Par volume	Trade Status	As Of	Msg Seq Number	Orig Msg Seq Number	Seller Commission
AIG.GYH	20121115	10:42:09	119	20,000	T		0022513		
AIG.GYH	20121115	10:42:09	119.31645	20,000	T	A	0051114		63.29
AIG.GYH	20121115	10:42:09	119	20,000	Y	R	0051115	0022513	

Table 4 Agency transaction.

The table contains a typical example of how a agency transaction. An introducing broker acts on behalf of a customer. The introducing broker transacts with an executing broker and hands on the bonds to the customer. The introducing broker is not allowed to make a markup or markdown in the price but can charge a commission. The variable in the table are bond ID (bond_sym_id), transaction date (trd_exctn_dt), transaction time (trd_exctn_tm), price (rptd_pr), par volume (entrd_vol_qt), Contra Party (contra_mp_id), Buy/Sell (rpt_side_cd), Buyer Capacity (buy_cpcty_cd), Seller Capacity (sell_cpcty_cd), and buyer commission (buy_cmsn_rt).

Bond ID	Transaction Date	Transaction Time	Price	Par volume	Contra Party	Buy/Sell	Buyer Capacity	Seller Capacity	Buyer Commission
AIG.GD	20071219	11:57:24	104.308	10,000	D	B	P	P	
AIG.GD	20071219	11:58:36	103.808	10,000	C	B	A	P	0.5
AIG.GD	20071219	11:58:36	104.308	10,000	D	S	P	A	

Table 5 Agency transaction prices.

This table shows average prices for transactions below $100,000. The prices have been estimated using differences between a target and a benchmark. For each bond on each day the benchmark is estimated as the average prices within that day for that bond. The difference is then between agency sells/buys and interdealer transactions, between agency sells/buys and non-agency customer sells/buys. Finally, average commission size have been added/substracted from the agency transactions. The estimation has been carried out for agency transactions with commission and without commission separately.

Transaction type	Price (w. comm)	Price (wo. comm)
Agency Sell + Commission	100.38	-
Customer Sell	100.65	100.44
Agency Sell	100.10	100.27
Interdealer	100	100
Agency Buy	99.67	99.67
Customer Buy	99.41	99.60
Agency Buy + Commission	99.17	-